Università degli Studi di Trieste
Dipartimento di Fisica Teorica

Cattedra di Fisica Terrestre e Climatologia

Franco Stravisi

A BASIC PROGRAM FOR PSYCHROMETRIC COMPUTATIONS AND MERCURY BAROMETER CORRECTIONS

Rapporto Interno
FTC 88/3

Trieste 1988

CONTENTS

A BASIC PROGRAM FOR PSYCHROMETRIC COMPUTATIONS AND MERCURY BAROMETER CORRECTIONS

Franco Stravisi
University of Trieste,
Department of Theoretical Physics

ABSTRACT. A BASIC program is given for (i) the computation of the atmospheric relative humidity by means of the dry and wet bulb temperatures given by the aspirating psychrometer and for (ii) the correction of the mercury Fortin barometer and the pressure reduction to the sea level.

1. Introduction

Daily routine work in a meteorological station requires frequent measurements of the relative humidity and of the atmospheric pressure for the calibration of the corresponding analogue or digital recorders. The aspirating psychrometer (Assmann or August type) and the mercury barometer (Fortin type) are conventionally used to this purpose. Percentual relative humidity values are usually found on tables or graphs as function of the psychrometric temperatures at a constant, close to the standard, pressure. Pressure readings on the barometer scale are reduced to the 0 °C scale and to the standard gravity by means of tables; the corresponding pressure at the sea level, for small station heights, is then usually computed by adding the monthly mean pressure of an air column of the same height.

The same kind of computations can be done on a computer by using the equation of state of the moist atmosphere and the corresponding Fortran programs described by Stravisi (1988). In this report a BASIC program is presented to perform the following operations:

(1) compute the main parameters concerning air humidity, given the psychrometric temperatures and an approximate value of the atmospheric pressure;

(2) compute the atmospheric pressure at the station height, given the pressure reading of a mercury barometer and its temperature;

(3) compute the atmospheric pressure reduction to the sea level, for a fixed station height, given the psychrometric temperatures (air temperature and humidity) or only the air temperature (approximate correction).

2. Psychrometric computations

The percentual relative humidity is commonly measured, in practical meteorology, by means of the aspirating psychrometer. This
instrument yields the dry (θ) and wet (θ_w) bulb Celsius temperatures. Sprung's psychrometric formula:

$$e(\theta) = e_w(\theta_w) - A\,p\,(\theta - \theta_w) \qquad (1)$$

relates the water vapour pressure e, the water vapour pressure at saturation e_w, the psychrometric temperature lapse and the atmospheric pressure p. The Ferrel's coefficient:

$$A = 0.000660\,(1 + 0.00115\,\theta_w/°C)\ \ °C^{-1} \qquad (2)$$

is used for ventilations in the range 4-10 m/s. The water vapour pressure at saturation as a function of the air temperature is given by the Goff-Gratch formula (Stravisi 1988); using the natural logarithm, this is:

$$L_1 = 10.79574\ T_2 - 2.183633 \log(T_3) + 0.78614\ ,$$
$$L_2 = 1.50475 \times 10^{-4} \left(1 - 10^{(-8.2969(T_3 - 1))}\right)\ ,$$
$$L_3 = 4.2873 \times 10^{-4} \left(10^{(4.76955\ T_2)} - 1\right)\ ;$$

$$e_w(T) = 10^{(L_1 + L_2 + L_3)}\ ; \qquad (3)$$

$$T = \theta + T_o\ ,$$
$$T_o = 273.15 \text{ K}\ ,$$
$$T_1 = 273.16 \text{ K}\ ,$$
$$T_2 = 1 - T_1/T\ ,$$
$$T_3 = T/T_1\ .$$

The moist atmosphere is characterized by p, θ, θ_w, ε, ε_w above and by the following quantities:

mole fraction	:	e/p
mixing ratio	:	$r = \varepsilon\, e/(p - e)$
virtual temperature	:	$T_v = T(1 + r/\varepsilon)/(1 + r)$
adjusted virtual temperature	:	$T_v' = 0.9995\ T_v$
moist air density	:	$\rho = p/(R\ T_v')$
specific humidity	:	$q = r/(1 + r)$
absolute humidity	:	$d_v = \rho\, q$
relative humidity	:	$u = (e/e_w)(p - e_w)/(p - e)$
percentual relative humidity	:	$U = 100\ u$ (4)

R = 287.053 $J.kg^{-1}.K^{-1}$ is the dry air constant and $\varepsilon = 0.62198$ is the molar mass ratio of water vapour and dry air. Further details are given by Stravisi (1988).

Given p, θ, θ_w, quantities above are computed by the BASIC program "PSYBAR" described in section 4.

3. The mercury barometer

The mercury barometer is the reference instrument for measuring the atmospheric pressure in a meteorological station. The atmospheric pressure is balanced by the pressure of the mercury column plus the pressure (p_c) resulting from the different curvature of the meniscus at the top of the mercury column and in the container:

$$p = \rho_m\, g\, h + p_c\ . \qquad (5)$$

The mercury pressure is a function of the local gravity g, of the mercury density ρ_m at the barometer temperature θ_b and of the column height h above the mercury level in the container. The thermal dilatation of mercury inside the barometer is linear, and:

$$\rho_m = \rho_o\, (1 - \mu\, \theta_b)\ , \qquad (6)$$

$$\mu = 1.818 \times 10^{-4}\ °C^{-1}\ .$$

Introducing the standard gravity g_o, (5,6) give:

$$p = (g/g_o)\,(1 - \mu\ \theta_b)\, \rho_o\, g_o\, h + p_c\ . \qquad (7)$$

Using SI units (IAPSO, 1979), the numerical values of the physical quantities in (7) are defined as follows:

$$\begin{aligned} P &= p/\text{Pa} , \\ R_o &= \rho_o/(\text{kg.m}^{-3}) = 13.5951 \times 10^3 , \\ G_o &= g_o/(\text{m.s}^{-2}) = 9.80665 , \\ R_oG_o &= 1.33322387 \times 10^5 , \\ H &= h/\text{m} , \end{aligned} \tag{8}$$

The equation between numerical values corresponding to (7) is:

$$P = (g/g_o)\,(1 - \mu\ \ \theta_b)\, R_o\, G_o\, H + P_c\,. \tag{9}$$

The mercury column height h is commonly measured on a scale drawn on the barometer, using a conventional unit of length. We can call the length unit at present in use for barometers the "barometer hectopascal", symbol σ. The barometer hectopascal increases with temperature according to the linear dilatation of the material which supports the scale drawings:

$$\sigma = \sigma_o\,(1 + \beta\ \ \theta_b)\,. \tag{10}$$

For a brass ruler it is:

$$\beta = 0.184 \times 10^{-4}\ °\text{C}^{-1}\,.$$

The barometer hectopascal at 0 °C is a constant unit of length; it is defined by:

$$1\ \sigma_o = 10^2\,(R_oG_o)^{-1}\ \text{m}\,. \tag{11}$$

The mercury column height is therefore

$$h = \text{H m} = \text{B}\ \sigma , \tag{12}$$

where B is its numerical value on the barometer scale, and

$$H = B\ \sigma/\text{m} = 10^2\, B\,(1 + \beta\ \ \theta_b)\,(R_oG_o)^{-1}\,. \tag{13}$$

Substituting (13) in (9):

$$10^{-2}\, P = (g/g_o)\,(1 - a\ \theta_b)\, B + 10^{-2}\, P_c , \tag{14}$$

where

$$a = \mu - \beta = 1.6339 \times 10^{-4}\ °\text{C}^{-1} \tag{15}$$

is the linear dilatation of mercury with respect to brass. The numerical value of the atmospheric pressure in hectopascals is therefore (14, 8, 9) given by:

$$p/\text{hPa} = (g/g_o)\,(1 - a\ \theta_b)\ h/\sigma + p_c/\text{hPa}\,. \tag{16}$$

Formula (16) is used to convert the barometer reading $B=h/\sigma$ in "barometer hectopascals" into the corresponding pressure value in hectopascals; local gravity, the capillary correction of the instrument and its temperature must be known. The reading reduction to 0 °C can be computed also as

$$(1\text{-a}\ \theta_b)\ B = B - C(\theta_b)\ ;$$

$C(q_b)$ are reported in **Table 1**.

If the barometer scale is in "millimeters", pressure values in (16) are substituted by the corresponding values expressed in "conventional millimiters of mercury". This unit, whose use is "strongly discouraged" (IAPSO, 1979), is defined by:

$$1 \text{ mmHg} = R_o G_o \times 10^{-5} \text{ hPa} = 1.333\ 223\ 87 \text{ hPa} .$$



Fig. 1. Fortin barometer.

Measurement of the mercury height. A reference point is used in the Fortin barometer to define h=0 (**Fig.1**); the mercury level is set to zero by adjusting the volume of the bulb. In the fixed-bulb barometers this operation is avoided, since the scale is compensated for zero level changes according to the surface ratio between capillary and bulb. A third method used in some barometers is taking the difference between the heights of the upper and lower mercury levels. If the bulb diameter is the same of the barometer tube at the reading level, the capillary corrections cancel each other. The correction for the capillary depression is given by:

$$p_c/\text{hPa} = 0.087 + 0.063\ X/\sigma ,$$

where X/σ is the meniscus height in barometer hectopascals.

Pressure reduction to sea level. The exact computation of the sea level pressure $p(0)$ requires the knowledge of the vertical profile of the air density up to the station height z. An approximate solution is usually given by assuming a constant air temperature for the atmospheric layer below z. Integration of the equation of state then yields:

$$p(0) = p(z) \exp(g\, z/(R\, T_v')) ; \qquad (17)$$

and, for typical atmospheric conditions,

$$p(0) = p(z) \exp(1.184 \times 10^{-4}\, z/\text{m}) . \qquad (18)$$

For a small station height (below about 100 m) a constant density can be assumed, and

$$p(0) = p(z) + \rho(z)\, g\, z . \qquad (19)$$

4. The BASIC program "PSYBAR"

The program list is on pages 8-9. Local gravity is defined on line `1010`; the standard value is used for Trieste. The capillary correction is defined on line `980`; p_c = 0.16 hPa is a default value for a Fortin barometer. The station height is defined on line `1000`: z = 31.8 m corresponds to the height of the Climatology Laboratory of the International Centre for Theoretical Physics at Grignano, Trieste. Values above can be changed for different instruments and situations.

Running the program, the first request is:

[1] "`Psychrometric computations only (Y/N) ?`": input "`Y`" or "`y`" or anything else. In the "`Y`" case, requests are:

[2] "`Atmospheric pressure /hPa ?`" (input p),
[3] "`Dry bulb temperature /°C ?`" (input θ),
[4] "`Wet bulb temperature /°C ?`" (input θ_w);

at this stage, PSYBAR computes and displays mole fraction, r, T_v, T_v', e, e_w, q, d_v, ρ and U. Control is then transferred to request [3], and so on.

If the answer to [1] is "N" (or different from "Y"), the following request is:
[5] "`Mercury barometer correction only (Y/N) ?`" If the answer is "Y", then follows:
[6] "`Air temperature ?`" (input θ). This is used for computing the pressure reduction to the sea level by means of an approximate air density (at 60 % relative humidity):

$$p(0) = p(z) + \rho\, g\, z, \qquad \rho = 1.2912\ (1 - 0.0035\ \theta/°\mathrm{C})\ \ \mathrm{kg/m^3}\ .$$

[7] "`Mercury barometer: pressure /hPa ?`" and
[8] " `temperature/°C ?`": input the reading B in barometer hectopascals and the barometer temperature θ_b. The
following results are displayed:

```
"Pressure at .... m:     ...... hPa",
"Correction to s.l.:     ...... hPa",
"Pressure at m.s.l.:     ...... hPa".
```

Control is then transferred again to [7].

If answers to [1] and to [5] are negative, the inputs are p, θ, θ_w [2, 3, 4]. Moist air quantities are computed and displayed as following [4]; B and θ_b are supplied following [7, 8], the correct barometric pressure and the correction to the sea level (19) are computed and displayed as after [8]. Control goes back to [3] and so on.

REFERENCES

IAPSO (1979): "SUN report on the use in Physical sciences of the ocean of the Systeme International d'Unites (SI) and related
standards for symbols and terminology", IUGG Publ. Off., France, 56 pp.

STRAVISI F. (1988): "Fortran programs for the equation of state of moist air", UTS-FTC 88/2, 15 pp.

```
10 REM- Program "PSYBAR"    (Franco Stravisi 1988, FTC 88/3)
20 REM- Psychrometric computations
30 REM- and mercury barometer corrections
40 REM- INPUT:  TA dry bulb air temperature /°C
50 REM-         TW wet bulb     temperature /°C
60 REM-         PA atmospheric pressure /hPa
70 REM- OUTPUT: NV mole fraction
80 REM-         R  mixing ratio x 1000
90 REM-         TV virtual temperature /K
100 REM-        TP adjusted virtual temperature /K
110 REM-        E  water vapour pressure /hPa
120 REM-        EW saturation vapour pressure /hPa
130 REM-        Q  specific humidity x 1000
140 REM-        D  absolute humidity /(g/m3)
150 REM-        U  relative humidity /%
160 REM-        RO moist air density /(kg/m3)
170 REM-------------------------------------------------
180 PRINT :PRINT
190 PRINT "------------------------------------------------"
200 PRINT "PSYCHROMETRIC COMPUTATIONS and"
210 PRINT "MERCURY BAROMETER  CORRECTIONS"
220 PRINT "(Franco Stravisi FTC 1988/3)"
230 PRINT
240 PRINT "The following constants are defined:"
250 PRINT "980 CM=0.16  barometer capillary corr. /hPa,"
260 PRINT "1000 Z=31.8  barometer height above sea level/m,"
270 PRINT "1010 G=9.80665  local gravity /(m.s-2) ."
280 PRINT "CHANGE VALUES ABOVE IF NECESSARY !"
290 PRINT "------------------------------------------------"
300 PRINT
310 INPUT "Psychrometric computations   only (Y/N)  "; P$
320 IF P$="y" THEN P$="Y"
330 IF P$="Y" GOTO 380
340 INPUT "Mercury barometer correction only (Y/N)  "; B$
350 IF B$="y" THEN B$="Y"
360 IF B$="Y" THEN PRINT "Approximate air density assumed !"
370 IF B$="Y" GOTO 900
380 INPUT "Atmospheric pressure /hPa  "; PA
390 PRINT
400 INPUT "Dry bulb temperature /°C   "; TA
410 INPUT "Wet bulb temperature /°C   "; TW
420 A=.00066*(1+.00115*TW)
430 T=TW
440 GOSUB 1190
450 EWW=EW
460 E=EWW-A*PA*(TA-TW)
470 NV=E/PA
480 T=TA
490 GOSUB 1190
500 EP=.62198
510 R=EP*E/(PA-E)
520 Q=1000*R/(R+1)
530 TV=(TA+273.15)*(1+R/EP)/(1+R)
540 TP=.9995*TV
550 RO=PA/(2.87053*TP)
560 D=RO*Q
570 U=100*E/EW*(PA-EW)/(PA-E)
580 PRINT
590 PRINT " Mole fraction      :";
600 PRINT USING "########.##"; NV*1000;
610 PRINT "    x 0.001"
620 PRINT " Mixing ratio       :";
630 PRINT USING "########.##"; R*1000;
640 PRINT "    x 0.001"
650 PRINT " Virtual temp.      :";
660 PRINT USING "########.##"; TV;
670 PRINT "    K"
680 PRINT " Adj. virt. temp.   :";
```

```
690 PRINT USING "########.##"; TP;
700 PRINT "    K"
710 PRINT " Vapor pressure     :";
720 PRINT USING "########.##"; E;
730 PRINT "    hPa"
740 PRINT " Saturation vap. pr.:";
750 PRINT USING "########.##"; EW;
760 PRINT "    hPa"
770 PRINT " Specific humidity  :";
780 PRINT USING "########.##"; Q;
790 PRINT "    x 0.001"
800 PRINT " Absolute humidity  :";
810 PRINT USING "########.##"; D;
820 PRINT "    g/m3"
830 PRINT " Moist air density  :";
840 PRINT USING "########.####"; RO;
850 PRINT "  kg/m3"
860 PRINT " Relative humidity  :";
870 PRINT USING "########.##"; U;
880 PRINT "    %"
890 IF P$="Y" GOTO 390
900 PRINT
910 IF B$<>"Y" GOTO 960
920 INPUT "Air temperature /°C  "; TA
930 REM- Compute approximate air density at 60%
940 RO=1.2912*(1-.00355*TA)
950 IF B$="Y" THEN PRINT
960 INPUT "Mercury barometer: pressure  /hPa"; PL
970 INPUT "                   temperature/°C"; TB
980 CM=.16
990 GS=9.80665
1000 Z=31.8
1010 G=9.80665
1020 ALFA=1.6339E-04
1030 P=(G/GS)*PL*(1-ALFA*TB)+CM
1040 CZ=RO*G*Z/100
1050 P0=P+CZ
1060 PRINT " Pressure at";
1070 PRINT USING "###.#";Z;
1080 PRINT " m :";
1090 PRINT USING "########.##"; P;
1100 PRINT "    hPa"
1110 PRINT " Correction to s.l. :";
1120 PRINT USING "########.##";CZ;
1130 PRINT "    hPa"
1140 PRINT " Pressure at m.s.l. :";
1150 PRINT USING "########.##"; P0;
1160 PRINT "    hPa"
1170 IF B$="Y" GOTO 950
1180 GOTO 390
1190 REM- SUBROUTINE FOR EW
1200 TK=T+273.15
1210 TT=273.16
1220 T2=1-TT/TK
1230 T3=TK/TT
1240 L1=10.79574*T2-2.183633*LOG(T3)+.78614
1250 L2=1.50475E-04*(1-10^(-8.2969*(T3-1)))
1260 L3=4.2873E-04*(10^(4.76955*T2)-1)
1270 L10EW=L1+L2+L3
1280 EW=10^L10EW
1290 RETURN
1300 END
```

°C	950	960	970	980	990	1000	1010	1020	1030	1040	1050
1	0.16	0.16	0.16	0.16	0.16	0.16	0.17	0.17	0.17	0.17	0.17
2	0.31	0.31	0.32	0.32	0.32	0.33	0.33	0.33	0.34	0.34	0.34
3	0.47	0.47	0.48	0.48	0.49	0.49	0.50	0.50	0.50	0.51	0.51
4	0.62	0.63	0.63	0.64	0.65	0.65	0.66	0.67	0.67	0.68	0.69
5	0.78	0.78	0.79	0.80	0.81	0.82	0.83	0.83	0.84	0.85	0.86
6	0.93	0.94	0.95	0.96	0.97	0.98	0.99	1.00	1.01	1.02	1.03
7	1.09	1.10	1.11	1.12	1.13	1.14	1.16	1.17	1.18	1.19	1.20
8	1.24	1.25	1.27	1.28	1.29	1.31	1.32	1.33	1.35	1.36	1.37
9	1.40	1.41	1.43	1.44	1.46	1.47	1.49	1.50	1.51	1.53	1.54
10	1.55	1.57	1.58	1.60	1.62	1.63	1.65	1.67	1.68	1.70	1.72
11	1.71	1.73	1.74	1.76	1.78	1.80	1.82	1.83	1.85	1.87	1.89
12	1.86	1.88	1.90	1.92	1.94	1.96	1.98	2.00	2.02	2.04	2.06
13	2.02	2.04	2.06	2.08	2.10	2.12	2.15	2.17	2.19	2.21	2.23
14	2.17	2.20	2.22	2.24	2.26	2.29	2.31	2.33	2.36	2.38	2.40
15	2.33	2.35	2.38	2.40	2.43	2.45	2.48	2.50	2.52	2.55	2.57
16	2.48	2.51	2.54	2.56	2.59	2.61	2.64	2.67	2.69	2.72	2.74
17	2.64	2.67	2.69	2.72	2.75	2.78	2.81	2.83	2.86	2.89	2.92
18	2.79	2.82	2.85	2.88	2.91	2.94	2.97	3.00	3.03	3.06	3.09
19	2.95	2.98	3.01	3.04	3.07	3.10	3.14	3.17	3.20	3.23	3.26
20	3.10	3.14	3.17	3.20	3.24	3.27	3.30	3.33	3.37	3.40	3.43
21	3.26	3.29	3.33	3.36	3.40	3.43	3.47	3.50	3.53	3.57	3.60
22	3.41	3.45	3.49	3.52	3.56	3.59	3.63	3.67	3.70	3.74	3.77
23	3.57	3.61	3.65	3.68	3.72	3.76	3.80	3.83	3.87	3.91	3.95
24	3.73	3.76	3.80	3.84	3.88	3.92	3.96	4.00	4.04	4.08	4.12
25	3.88	3.92	3.96	4.00	4.04	4.08	4.13	4.17	4.21	4.25	4.29
26	4.04	4.08	4.12	4.16	4.21	4.25	4.29	4.33	4.38	4.42	4.46
27	4.19	4.24	4.28	4.32	4.37	4.41	4.46	4.50	4.54	4.59	4.63
28	4.35	4.39	4.44	4.48	4.53	4.57	4.62	4.67	4.71	4.76	4.80
29	4.50	4.55	4.60	4.64	4.69	4.74	4.79	4.83	4.88	4.93	4.98
30	4.66	4.71	4.75	4.80	4.85	4.90	4.95	5.00	5.05	5.10	5.15
31	4.81	4.86	4.91	4.96	5.01	5.07	5.12	5.17	5.22	5.27	5.32
32	4.97	5.02	5.07	5.12	5.18	5.23	5.28	5.33	5.39	5.44	5.49
33	5.12	5.18	5.23	5.28	5.34	5.39	5.45	5.50	5.55	5.61	5.66
34	5.28	5.33	5.39	5.44	5.50	5.56	5.61	5.67	5.72	5.78	5.83
35	5.43	5.49	5.55	5.60	5.66	5.72	5.78	5.83	5.89	5.95	6.00
36	5.59	5.65	5.71	5.76	5.82	5.88	5.94	6.00	6.06	6.12	6.18
37	5.74	5.80	5.86	5.92	5.98	6.05	6.11	6.17	6.23	6.29	6.35
38	5.90	5.96	6.02	6.08	6.15	6.21	6.27	6.33	6.40	6.46	6.52
39	6.05	6.12	6.18	6.24	6.31	6.37	6.44	6.50	6.56	6.63	6.69
40	6.21	6.27	6.34	6.40	6.47	6.54	6.60	6.67	6.73	6.80	6.86

TABLE 1. Corrections to reduce mercury barometer readings (950-1050) at given instrument temperature (1-40 °C) to the 0 °C scale.